1.5 Střed, singulární body kvadriky

```
> restart;
> with(LinearAlgebra):
> X:=Vector[row]([x,y,z,1]);
```

$$X := [x, y, z, 1]$$

Matice kvadriky:

```
> K:=Matrix(a,1..4,1..4,shape=symmetric);
```

$$K := \begin{bmatrix} a(1,1) & a(1,2) & a(1,3) & a(1,4) \\ a(1,2) & a(2,2) & a(2,3) & a(2,4) \\ a(1,3) & a(2,3) & a(3,3) & a(3,4) \\ a(1,4) & a(2,4) & a(3,4) & a(4,4) \end{bmatrix}$$

Rovnice kvadriky:

```
> Kv:=expand(evalm(X&*K&*Transpose(X)))=0;
```

$$Kv := x^2\, a(1,1) + 2\, x\, y\, a(1,2) + 2\, x\, z\, a(1,3) + 2\, x\, a(1,4) + y^2\, a(2,2) + 2\, y\, z\, a(2,3) + 2\, y\, a(2,4) + z^2\, a(3,3) + 2\, z\, a(3,4) + a(4,4) = 0$$

Bod přímky:

```
> M:=[m,n,p];
```

$$M := [m, n, p]$$

Parametrické rovnice přímky:

```
> Primka:=[x=m+t*u,y=n+t*v,z=p+t*w];
```

$$Primka := [x = m + t\, u,\ y = n + t\, v,\ z = p + t\, w]$$

Parametrické rovnice přímky dosazené do rovnice kvadriky:

```
> Kv1:=simplify(eval(Kv,Primka));
```

$$Kv1 := 2\, a(1,1)\, m\, t\, u + 2\, a(1,2)\, m\, t\, v + 2\, a(1,2)\, t\, u\, n + 2\, a(1,2)\, t^2\, u\, v + 2\, a(1,3)\, m\, t\, w + 2\, a(1,3)\, t\, u\, p + 2\, a(1,3)\, t^2\, u\, w + 2\, a(2,2)\, n\, t\, v + 2\, a(2,3)\, n\, t\, w + 2\, a(2,3)\, t\, v\, p + 2\, a(2,3)\, t^2\, v\, w + 2\, a(3,3)\, p\, t\, w + a(1,1)\, m^2 + 2\, a(1,4)\, m + a(2,2)\, n^2 + 2\, a(2,4)\, n + a(3,3)\, p^2 + 2\, a(3,4)\, p + a(1,1)\, t^2\, u^2 + 2\, a(1,2)\, m\, n + 2\, a(1,3)\, m\, p + 2\, a(1,4)\, t\, u + a(2,2)\, t^2\, v^2 + 2\, a(2,3)\, n\, p + 2\, a(2,4)\, t\, v + a(3,3)\, t^2\, w^2 + 2\, a(3,4)\, t\, w + a(4,4) = 0$$

Koeficient B rovnice $At^2 + Bt + C = 0$ společných bodů kvadriky a přímky:

```
> B:=coeff(lhs(Kv1),t)/2;
```

$$B := a(1,1)\, m\, u + a(1,2)\, m\, v + a(1,2)\, u\, n + a(1,3)\, m\, w + a(1,3)\, u\, p + a(2,2)\, n\, v + a(2,3)\, n\, w + a(2,3)\, v\, p + a(3,3)\, p\, w + a(1,4)\, u + a(2,4)\, v + a(3,4)\, w$$

```
> B:=collect(B,[u,v,w]);
```

$$B := (a(1,1)\, m + a(1,3)\, p + a(1,4) + a(1,2)\, n)\, u + (a(1,2)\, m + a(2,2)\, n + a(2,4) + a(2,3)\, p)\, v$$

$$+(a(3,3)\,p + a(2,3)\,n + a(1,3)\,m + a(3,4))\,w$$

Soustava rovnic - podmínek pro určení bodu $M = [m, n, p]$ jako středu kvadriky :

```
> rS1:=sort(coeff(B,u),[m,n,p])=0;
  rS2:=sort(coeff(B,v),[m,n,p])=0;
  rS3:=sort(coeff(B,w),[m,n,p])=0;
```

$$rS1 := a(1,1)\,m + a(1,2)\,n + a(1,3)\,p + a(1,4) = 0$$

$$rS2 := a(1,2)\,m + a(2,2)\,n + a(2,3)\,p + a(2,4) = 0$$

$$rS3 := a(1,3)\,m + a(2,3)\,n + a(3,3)\,p + a(3,4) = 0$$

Determinant této soustavy - tzv. **hlavní minor** kvadriky:

```
> A44:=SubMatrix(K,[1..3],[1..3]);
```

$$A44 := \begin{bmatrix} a(1,1) & a(1,2) & a(1,3) \\ a(1,2) & a(2,2) & a(2,3) \\ a(1,3) & a(2,3) & a(3,3) \end{bmatrix}$$

Příklad: Napište rovnici kvadriky, která prochází bodem A, má střed S a protíná rovinu $z = 0$ v dané kuželosečce.

```
> A:=[2,0,-1]; S:=[0,0,-1]; kz:=x^2-4*x*y-1=0;
```

$$A := [2, 0, -1]$$

$$S := [0, 0, -1]$$

$$kz := x^2 - 4\,x\,y - 1 = 0$$

```
> rS1:=eval(rS1,{m=S[1],n=S[2],p=S[3]});
  rS2:=eval(rS2,{m=S[1],n=S[2],p=S[3]});
  rS3:=eval(rS3,{m=S[1],n=S[2],p=S[3]});
```

$$rS1 := -a(1,3) + a(1,4) = 0$$

$$rS2 := -a(2,3) + a(2,4) = 0$$

$$rS3 := -a(3,3) + a(3,4) = 0$$

```
> res1:=solve({rS1,rS2,rS3},{a(1,3),a(2,3),a(3,3)});
```

$$res1 := \{a(1,3) = a(1,4),\ a(2,3) = a(2,4),\ a(3,3) = a(3,4)\}$$

```
> assign(res1);
> K1:=K;
```

$$K1 := \begin{bmatrix} a(1,1) & a(1,2) & a(1,4) & a(1,4) \\ a(1,2) & a(2,2) & a(2,4) & a(2,4) \\ a(1,4) & a(2,4) & a(3,4) & a(3,4) \\ a(1,4) & a(2,4) & a(3,4) & a(4,4) \end{bmatrix}$$

```
> z:=0;
```

$$z := 0$$

```
> Kv; kz;
```

$$x^2\,a(1,1) + 2\,x\,y\,a(1,2) + 2\,x\,a(1,4) + y^2\,a(2,2) + 2\,y\,a(2,4) + a(4,4) = 0$$

$$x^2 - 4\,x\,y - 1 = 0$$

```
> res2:=solve({coeffs(lhs(Kv)-lhs(kz))},{a(1,1),a(1,2),a(1,4),a(2,
  2),a(2,4),a(4,4)});
```

$$res2 := \{a(1,1)=1, a(1,2)=-2, a(1,4)=0, a(2,2)=0, a(2,4)=0, a(4,4)=-1\}$$

```
> assign(res2);
> K2:=map(x->eval(x),K1);
```

$$K2 := \begin{bmatrix} 1 & -2 & 0 & 0 \\ -2 & 0 & 0 & 0 \\ 0 & 0 & a(3,4) & a(3,4) \\ 0 & 0 & a(3,4) & -1 \end{bmatrix}$$

```
> z:='z';
```

$$z := z$$

```
> Kv:=eval(Kv);
```

$$Kv := x^2 - 4\,x\,y - 1 + z^2\,a(3,4) + 2\,z\,a(3,4) = 0$$

```
> res3:=solve(eval(Kv,[x=2,y=0,z=-1]),{a(3,4)});
```

$$res3 := \{a(3,4) = 3\}$$

```
> assign(res3);
> Kv:=eval(Kv);
```

$$Kv := x^2 - 4\,x\,y - 1 + 3\,z^2 + 6\,z = 0$$

```
> A44:=map(x->eval(x),A44);
```

$$A44 := \begin{bmatrix} 1 & -2 & 0 \\ -2 & 0 & 0 \\ 0 & 0 & 3 \end{bmatrix}$$

```
> Determinant(A44);
```

$$-12$$

Hlavní minor je různý od nuly. Kvadrika je středová.

```
> plotsetup(inline,plotoptions=`portrait,noborder,shrinkby=0`);
> plots[implicitplot3d](Kv,x=-4..4,y=-4..4,z=-4..4,axes=frame,orie
ntation=[-50,70],color=COLOR(RGB,205/255,205/255,205/255),style=
patchcontour,grid=[30,30,40],light=[90,-5,1,1,1],tickmarks=[3,3,
3],scaling=constrained,view=[-3..3,-3..3,-3..3]);
```




```
> Plocha_patch:=plots[implicitplot3d](Kv,x=-4..4,y=-4..4,z=-4..4,color=COLOR(RGB,205/255,205/255,205/255),style=patchcontour,grid=[30,30,40],light=[90,-5,1,1,1],scaling=constrained):
> Plocha:=plots[implicitplot3d](Kv,x=-5..5,y=-5..5,z=-5..5,color=COLOR(RGB,200/255,200/255,200/255),style=wireframe,grid=[30,30,30],scaling=constrained):
> PlochaColor:=plots[implicitplot3d](Kv,x=-8..8,y=-8..8,z=-5..5,style=wireframe,grid=[30,30,30],scaling=constrained):
> Krivka1:=plot3d([t,solve(eval(kz,x=t),y),0],t=-3..0,s=-3..3,thickness=3,color=red):
> Krivka2:=plot3d([t,solve(eval(kz,x=t),y),0],t=0..3,s=-3..3,thickness=3,color=red):
> Bod:=plottools[sphere]([2,0,-1],0.08,color=red):
> Stred:=plottools[sphere]([0,0,-1],0.08,color=red):
> Kvadr:=plottools[cuboid]([0,0,0],[2,0,-1],style=wireframe,color=grey):
> Rov_z:=plot3d(0,x=-6..6,y=-6..6,style=wireframe,color=grey):
> plots[display](Krivka1,Krivka2,Bod,
Kvadr,Stred,Rov_z,axes=frame,orientation=[62,75],tickmarks=[3,2,2],scaling=constrained,view=[-3..3,-2..2,-2..2],light=[90,-20,1,1,1]);
```




```
> plots[display](Krivka1,Krivka2,Bod,
  Kvadr,Plocha_patch,Stred,Rov_z,axes=frame,orientation=[62,75],ti
  ckmarks=[3,3,3],scaling=constrained,view=[-3..3,-3..3,-3..3],lig
  ht=[90,-20,1,1,1]);
```




```
>
```